#include <Wire.h>
#include <Adafruit_GFX.h>
#include <Adafruit_SSD1306.h>
#include <DHT.h>

// OLED显示屏配置
#define SCREEN_WIDTH 128
#define SCREEN_HEIGHT 64
#define OLED_RESET -1
#define SCREEN_ADDRESS 0x3C

// IIC引脚定义
#define SDA_PIN 7
#define SCL_PIN 6

// DHT11引脚定义
#define DHTPIN 10
#define DHTTYPE DHT11

// 创建对象
Adafruit_SSD1306 display(SCREEN_WIDTH, SCREEN_HEIGHT, &Wire, OLED_RESET);
DHT dht(DHTPIN, DHTTYPE);

void setup() {
  Serial.begin(115200);
  
  // 初始化IIC
  Wire.begin(SDA_PIN, SCL_PIN);
  
  // 初始化OLED
  if(!display.begin(SSD1306_SWITCHCAPVCC, SCREEN_ADDRESS)) {
    Serial.println(F("SSD1306 allocation failed"));
    for(;;);
  }
  
  // 清空屏幕
  display.clearDisplay();
  display.setTextColor(SSD1306_WHITE);
  
  // 显示启动画面
  display.setTextSize(2);
  display.setCursor(20, 20);
  display.println(F("iCEasy"));
  display.setTextSize(1);
  display.setCursor(10, 45);
  display.println(F("Initializing..."));
  display.display();
  
  // 初始化DHT11
  dht.begin();
  
  delay(2000);
  
  Serial.println(F("System Ready!"));
}

void loop() {
  // 读取温湿度数据
  float humidity = dht.readHumidity();
  float temperature = dht.readTemperature();
  
  // 检查读取是否成功
  if (isnan(humidity) || isnan(temperature)) {
    Serial.println(F("Failed to read from DHT sensor!"));
    displayError();
    delay(2000);
    return;
  }
  
  // 串口输出
  Serial.print(F("Temperature: "));
  Serial.print(temperature);
  Serial.print(F("°C  Humidity: "));
  Serial.print(humidity);
  Serial.println(F("%"));
  
  // 显示在OLED上
  displayData(temperature, humidity);
  
  delay(2000);  // 每2秒更新一次
}

void displayData(float temp, float humi) {
  display.clearDisplay();
  
  // 第一行：显示iCEasy标题
  display.setTextSize(1);
  display.setCursor(0, 0);
  display.println(F("===== iCEasy ====="));
  
  // 显示分割线
  display.drawLine(0, 12, 128, 12, SSD1306_WHITE);
  
  // 显示温度
  display.setTextSize(1);
  display.setCursor(0, 20);
  display.print(F("Temperature:"));
  
  display.setTextSize(2);
  display.setCursor(10, 32);
  display.print(temp, 1);
  display.setTextSize(1);
  display.setCursor(70, 35);
  display.print(F("'C"));
  
  // 显示湿度
  display.setTextSize(1);
  display.setCursor(0, 50);
  display.print(F("Humidity:"));
  display.setTextSize(1);
  display.setCursor(70, 50);
  display.print(humi, 1);
  display.print(F("%"));
  
  display.display();
}

void displayError() {
  display.clearDisplay();
  
  // 第一行：显示iCEasy标题
  display.setTextSize(1);
  display.setCursor(0, 0);
  display.println(F("===== iCEasy ====="));
  
  display.drawLine(0, 12, 128, 12, SSD1306_WHITE);
  
  display.setTextSize(1);
  display.setCursor(15, 30);
  display.println(F("Sensor Error!"));
  display.setCursor(10, 45);
  display.println(F("Check DHT11..."));
  
  display.display();
}

